Univ.-Prof. Dr.-Ing. habil. J. BETTEN, RWTH Aachen
Kabel (Freileitung) zwischen zwei Masten:

```
> restart:
> Dgl:=diff(y(x),x$2)-a*sqrt(1+(diff(y(x),x))^2)=0;
```

$$Dgl := \left(\frac{d^2}{dx^2}\,y(x)\right) - a\sqrt{1+\left(\frac{d}{dx}\,y(x)\right)^2} = 0$$

Darin ist der Parameter a gemäß a:= q / H definiert, wobei q = q(x,y) die auf die Längeneinheit des Kabels bezogene Kraftkomponente (z.B.: Eigengewicht + Windkraft + Schnee- & Eislast) ist, während H die auf die Stützen wirkende Horizontalkraft bedeutet. Bei gegebenem q = q(x,y) kann H aus der Stabilität der Stützen vorgeschrieben werden, so dass damit der Parameter a bestimmt ist, der die Länge des Kabels festlegt. Mithin kann für ein Kabel mit gegebenem q = q(x,y) die erforderliche Kabellänge allein über den Horizontalzug bestimmt werden. Zu berücksichtigen ist auch der Temperatureinfluss (Sommer / Winter) auf die Kabeldehnung.
Mit abnehmendem Horizontalzug H nimmt der Parameter a und damit auch die Kabellänge zu, so dass die vertikale Belastung der Stützen größer wird.

Randbedingungen # Aufhängung in Hanglage:

```
> R[1]:=y(0)=1;   R[2]:=y(2)=2;  #  beispielsweise
```

$$R_1 := y(0) = 1$$

$$R_2 := y(2) = 2$$

```
> dsolve({Dgl,R[1],R[2]},y(x)):
> Y(x,a):=convert(rhs(%),radical):
> Ableitung(x,a):=simplify(diff(%,x))  assuming a>0;
```

$$\mathrm{Ableitung}(x,a) := \sinh\left(x\,a - a + \ln\left(a\,\mathbf{e}^a + \sqrt{a^2\,\mathbf{e}^{(2a)} - 2\,\mathbf{e}^{(2a)} + \mathbf{e}^{(4a)} + 1}\right) - \ln\left(\mathbf{e}^{(2a)} - 1\right)\right)$$

```
> for i in [1,1.5,2,2.5] do Y(x)[a=i]:=
  evalf(subs(a=i,Y(x,a)))   od;
```

$$Y(x)_{a=1} := \cosh(x - 0.5864315493) - 0.1769356748$$

$$Y(x)_{a=1.5} := 0.6666666667\,\cosh(1.5\,x - 1.154672647) - 0.1627150844$$

$$Y(x)_{a=2} := 0.5000000000\,\cosh(2.\,x - 1.727658531) - 0.4512904872$$

$$Y(x)_{a=2.5} := 0.4000000000\,\cosh(2.5\,x - 2.294837712) - 1.004720654$$

```
> for i in [1,1.5,2,2.5] do
  Neigung(x)[a=i]:=evalf(subs(a=i,diff(Y(x,a),x))) od:
> erforderliche_Kabellänge:=Int(sqrt(1+(diff(y(x),x))^2),x=0..2)=
  value(int(sqrt(1+(sinh(a*x-a+ln(a*exp(a)+sqrt(a^2*exp(2*a)-2*exp
  (2*a)+ exp(4*a)+1))-ln(exp(2*a)-1)))^2),x=0..2))  assuming a>0:
> alias(H=Heaviside,th=thickness):
> plot1:=plot(rhs(%%),a=0..2.5,2..5,color=black,th=2):
> plot2:=plot({5,5*H(a-2.5)},a=0..2.5001,color=black,
  title="erforderliche Kabellänge  L(a)"):
```

```
> plots[display]({plot1,plot2});
```




```
> for i in [1,1.5,2,2.5] do erforderliche_Kabellänge[a=i]:=
  Int(sqrt(1+(diff(y(x),x))^2),x=0..2)=
  int(sqrt(1+(Neigung(x)[a=i])^2),x=0..2)  od;
```

$$erforderliche\_Kabellänge_{a=1} := \int_0^2 \sqrt{1+\left(\frac{d}{dx}\mathrm{y}(x)\right)^2}\,dx = 2.554288821$$

$$erforderliche\_Kabellänge_{a=1.5} := \int_0^2 \sqrt{1+\left(\frac{d}{dx}\mathrm{y}(x)\right)^2}\,dx = 3.010007308$$

$$erforderliche\_Kabellänge_{a=2} := \int_0^2 \sqrt{1+\left(\frac{d}{dx}\mathrm{y}(x)\right)^2}\,dx = 3.762195691$$

$$erforderliche\_Kabellänge_{a=2.5} := \int_0^2 \sqrt{1+\left(\frac{d}{dx}\mathrm{y}(x)\right)^2}\,dx = 4.942386420$$

```
> alias(H=Heaviside,th=thickness):
> plot1:=plot({Y(x)[a=1],Y(x)[a=1.5],Y(x)[a=2],Y(x)[a=2.5]},
  x=0..2,-1..2,scaling=constrained,th=2,color=black):
> plot2:=plot(2*H(x-2),x=0..2.001,color=black,
  title="Kabel zwischen zwei Masten # Parameter a = [1, 1.5,
  2, 2.5]"):
> plots[display]({plot1,plot2});
```




```
>
```

Symmetrische Aufhängung im Abstand von x = 1:

```
> restart:
> Dgl:=diff(y(x),x$2)-a*sqrt(1+(diff(y(x),x))^2)=0;
```

$$Dgl := \left(\frac{d^2}{dx^2}\,\mathrm{y}(x)\right) - a\sqrt{1+\left(\frac{d}{dx}\,\mathrm{y}(x)\right)^2} = 0$$

```
> R[1]:=y(0)=0;    R[2]:=y(1)=0;
```

$$R_1 := \mathrm{y}(0) = 0$$

$$R_2 := \mathrm{y}(1) = 0$$

```
> dsolve({Dgl,R[1],R[2]},y(x));
```

$$\mathrm{y}(x) = \frac{\cosh(x\,a + \ln(\mathrm{RootOf}(\mathbf{e}^a\,\_Z^2 - 1)))}{a} - \frac{1}{2}\,\frac{(\mathbf{e}^a + 1)\,\mathrm{RootOf}(\mathbf{e}^a\,\_Z^2 - 1)}{a}$$

```
> Y(x,a):=convert(rhs(%),radical);
```

$$\mathrm{Y}(x, a) := \frac{\cosh\left(x\,a + \ln\left(\sqrt{\frac{1}{\mathbf{e}^a}}\right)\right)}{a} - \frac{1}{2}\,\frac{(\mathbf{e}^a + 1)\sqrt{\frac{1}{\mathbf{e}^a}}}{a}$$

```
> Ableitung(x,a):=diff(%,x)  assuming a>0;
```

$$\mathrm{Ableitung}(x, a) := \sinh\left(x\,a - \frac{1}{2}\,a\right)$$

```
> for i in [1,1.5,2,2.5] do Y(x)[a=i]:=
  evalf(subs(a=i,Y(x,a))) od;
```

$$\mathrm{Y}(x)_{a=1} := \cosh(x - 0.5000000000) - 1.127625965$$

$$\mathrm{Y}(x)_{a=1.5} := 0.6666666667\,\cosh(1.5\,x - 0.7499999999) - 0.8631221902$$

$$\mathrm{Y}(x)_{a=2} := 0.5000000000\,\cosh(2.\,x - 1.000000000) - 0.7715403172$$

$$\mathrm{Y}(x)_{a=2.5} := 0.4000000000\,\cosh(2.5\,x - 1.250000000) - 0.7553695510$$

```
> for i in [1,1.5,2,2.5] do Neigung(x)[a=i]:=
  evalf(subs(a=i,diff(Y(x,a),x))) od;
```

$$\mathrm{Neigung}(x)_{a=1} := \sinh(x - 0.5000000000)$$

$$\mathrm{Neigung}(x)_{a=1.5} := \sinh(1.5\,x - 0.7499999999)$$

$$\mathrm{Neigung}(x)_{a=2} := \sinh(2.\,x - 1.000000000)$$

$$\mathrm{Neigung}(x)_{a=2.5} := \sinh(2.5\,x - 1.250000000)$$

```
> erforderliche_Kabellänge:= Int(sqrt(1+(diff(y(x),x))^2),x=0..1)=
  value(int(sqrt(1+(sinh(a*x-a/2))^2),x=0..1))  assuming a>0;
```

$$erforderliche\_Kabellänge := \int_0^1 \sqrt{1 + \left(\frac{d}{dx}\,\mathrm{y}(x)\right)^2}\,dx = \frac{\mathbf{e}^{\left(-\frac{a}{2}\right)}\,(-1 + \mathbf{e}^a)}{a}$$

```
> alias(H=Heaviside,th=thickness):
> plot1:=plot((exp(-a/2))*(-1+exp(a))/a,a=0..3,1..1.421,
  color=black,th=2):
> plot2:=plot({1.42,1.42*H(a-3)},a=0..3.001,color=black,
  title="erforderliche Kabellände  L(a)"):
> plots[display]({plot1,plot2});
```




```
> for i in [1,1.5,2,2.5] do erforderliche_Kabellänge[a=i]:=
  Int(sqrt(1+(diff(y(x),x))^2),x=0..1)=
  int(sqrt(1+(Neigung(x)[a=i])^2),x=0..1)  od;
```

$$erforderliche\_Kabellänge_{a=1} := \int_0^1 \sqrt{1 + \left(\frac{d}{dx}\,\mathrm{y}(x)\right)^2}\,dx = 1.042190611$$

$$erforderliche\_Kabellänge_{a=1.5} := \int_0^1 \sqrt{1 + \left(\frac{d}{dx}\,\mathrm{y}(x)\right)^2}\,dx = 1.096422309$$

$$erforderliche\_Kabellänge_{a=2} := \int_0^1 \sqrt{1 + \left(\frac{d}{dx}\,\mathrm{y}(x)\right)^2}\,dx = 1.175201194$$

$$erforderliche\_Kabellänge_{a=2.5} := \int_0^1 \sqrt{1 + \left(\frac{d}{dx}\,\mathrm{y}(x)\right)^2}\,dx = 1.281535264$$

```
> alias(H=Heaviside,th=thickness):
```

```
> plot1:=plot({Y(x)[a=1],Y(x)[a=1.5],Y(x)[a=2],Y(x)[a=2.5]},
  x=0..1,scaling=constrained,th=2,color=black):
> plot2:=plot({-0.5,-0.5*H(x-1)},x=0..1.001,color=black,
  title="Parameter  a  =  [1,  1.5,  2,  2.5]"):
> plots[display]({plot1,plot2});
```




```
> Y(1/2,a):=subs(x=1/2,Y(x,a));
```

$$\mathrm{Y}\left(\frac{1}{2}, a\right) := \frac{\cosh\left(\frac{a}{2} + \ln\left(\sqrt{\frac{1}{\mathbf{e}^a}}\right)\right)}{a} - \frac{1}{2}\,\frac{(\mathbf{e}^a + 1)\sqrt{\frac{1}{\mathbf{e}^a}}}{a}$$

```
> for i in [1,1.5,2,2.5] do maximaler_Durchhang[a=i]:=
  evalf(subs(a=i,Y(1/2,a)))  od;
```

$$maximaler\_Durchhang_{a=1} := -0.127625965$$

$$maximaler\_Durchhang_{a=1.5} := -0.1964555235$$

$$maximaler\_Durchhang_{a=2} := -0.2715403172$$

$$maximaler\_Durchhang_{a=2.5} := -0.3553695510$$

```
> maximaler_Durchhang[a=0]:=Limit(Y(1/2,a),a=0);
```

$$maximaler\_Durchhang_{a=0} := \lim_{a \to 0} \frac{\cosh\left(\frac{a}{2} + \ln\left(\sqrt{\frac{1}{\mathbf{e}^a}}\right)\right)}{a} - \frac{1}{2}\,\frac{(\mathbf{e}^a + 1)\sqrt{\frac{1}{\mathbf{e}^a}}}{a}$$

```
> maximaler_Durchhang[a=0]:=value(%);
```

$$maximaler\_Durchhang_{a=0} := 0$$

Für $a = 0$ verschwindet der Durchhang. Dann wird der Horizontalzug wegen $a = q / H$ unendlich groß.

```
>
```